

## 9/3.2

# Renversé

Ce jeu de réflexion demande beaucoup de pratique pour être utilisé de façon optimale.

Le principe du jeu est simple :  $n$  lettres ( $n$  compris entre 2 et 26) sont affichées à l'écran dans le désordre ; il faut les afficher dans l'ordre alphabétique en retournant successivement les  $x$  premières ( $x$  compris entre 2 et  $n$ ). Le jeu se décompose en deux phases :

1°) *Initialisation* :

- Choix de la dimension du jeu,
- Tirage aléatoire des lettres.

2°) *Jeu* :

- Affichage des lettres,
- Prise en compte de l'action du joueur,
- Test sur la fin du jeu.

Le programme écrit en BASIC est le suivant :

```

1000 REM =====|-----|-----|-----|-----|-----|-----|
1010 REM                               Jeu du RENVERSE
1020 REM =====
1030 GOSUB 2000 'Initialisation
1040 GOSUB 3000 'Jeu
1050 END

2000 REM =====
2010 REM                               Initialisation
2020 REM =====
2030 MODE 1:PRINT"JEU DU RENVERSE":PRINT
2040 DIM T$(26),T2$(26) 'Tableaux de jeu
2050 PRINT:INPUT"Dimension du jeu (entre 2 et 26) I"ID

```

```
2060 '
2070 'Tirage aleatoire du jeu
2080 '
2090 FOR I=1 TO D
2100   A=65+INT(RND(1)*D) 'Une lettre comprise entre A et chr$(65+D)
2110   N=0 'Indicateur de tirage deja fait
2120   FOR J=1 TO I
2130     IF T$(J)=CHR$(A) THEN N=1
2140   NEXT J
2150   IF N=1 THEN 2100 'Tirage deja fait
2160   T$(I)=CHR$(A)
2170 NEXT I
2180 RETURN

3000 REM =====
3010 REM                               Jeu
3020 REM =====
3030 '
3040 'Affichage du jeu
3050 '
3060 CLS:LOCATE 1,10
3070 FOR I=1 TO D
3080   PRINT T$(I);" ";
3090 NEXT I
3100 IF FIN=1 THEN LOCATE 1,16:PRINT"Jeu reussi en";C;"coups":RETURN
3110 '
3120 'Action du joueur
3130 '
3140 C=C+1 'Nombre de coups + 1
3150 LOCATE 1,20:PRINT"Nombre a retourner"
3160 INPUT"(a partir de la gauche)";NR
3170 IF NR>D THEN SOUND 1,100,30:GOTO 3150
3180 '
```

```
3190 'Prise en compte de l'action
3200 '
3210 FOR I=1 TO NR
3220   T$(I)=T$(NR+1-I)
3230 NEXT I
3240 FOR I=1 TO NR
3250   T$(I)=T2$(I)
3260 NEXT I
3270 '
3280 'Test de la fin du jeu
3290 '
3300 BP=0 'Nombre de lettre(s) bien placee(s)
3310 FOR I=1 TO D
3320   IF T$(I)=CHR$(64+I) THEN BP=BP+1
3330 NEXT I
3340 IF BP=D THEN FIN=1
3350 GOTO 3030 'Boucle de jeu
```

Lignes 1000 à 1050 : Programme principal.

Lignes 2000 à 2180 : Initialisation.

Lignes 3000 à 3350 : Jeu.

